1984/35. SPINEL: a FORTRAN program to correct electron probe spinel analyses for ferric oxide.

J.W. Hudspeth
A.V. Brown

Abstract

Electron probe analyses of spinel are input, are corrected for Fe_2O_3, and the structural formulae and percentage weights recalculated.

PROGRAM SPINEL

Where electron microprobe analyses return total iron reported as FeO, but in fact there exists also Fe_2O_3, then there is a need to re-compute the results in order to take this into account. The program was written to follow steps of a method provided by A.V. Brown (Appendix 1). The program is written in FORTRAN and runs interactively on the Department of Mines Perkin-Elmer 8/32 computer. The user is required to type "SPINEL" (which starts the .CSS of that name) and then to respond to a series of prompts. An example of an interactive run, together with the printed output obtained, is given in Appendix 2. Appendix 3 is a program listing. It should be noted that the analysis is made available (for program input) in oxide weight form where the assumption that the iron is in the form FeO has been applied. The method (Appendix 1) thus first reverses this step to cation proportions. The program also calculates and prints the Mg to Mg+Fe ratio.

[29 May 1984]

APPENDIX 1

Method used by program

Program is required for calculating a theoretical amount of Fe_2O_3 in a spinel structure using data obtained from electron probe, and assuming:

(1) Minerals of the spinel group have 32 oxygen ions and 24 cations in a unit cell.

(2) General formula is $R_8^{2+}\ R_{16}^{3+}\ O_{32}$.

(3) Assuming the general formula, apportions total iron as Fe^{2+} + Fe^{3+} such that 2 $(Mg+Mn+Ca+Fe^{2+})$ = $(Cr+Al+Ti+Fe^{3+})$.

INPUT

List the composition of the mineral expressed as weight percent of the constituent oxides as obtained from the probe.

List of possible oxides in standard order:

TiO_2
Al_2O_3
Cr_2O_3
FeO
MnO
NiO
MgO
CaO

CALCULATION

Step 1. Divide each oxide by its molecular weight (thus obtaining the molecular proportion of each oxide) and multiply the molecular proportion by the number of cations in the oxide concerned (this gives a set of numbers proportioned to the number of oxygen atoms associated with each oxide).

i.e.
TiO_2 ÷ 79.90 x 1 = X_1
Al_2O_3 ÷ 101.94 x 2 = X_2
Cr_2O_3 ÷ 152.02 x 2 = X_3
FeO ÷ 71.85 x 1 = X_4
MnO ÷ 70.94 x 1 = X_5
NiO ÷ 74.71 x 1 = X_6
MgO ÷ 40.32 x 1 = X_7
CaO ÷ 56.08 x 1 = X_8

Step 2. Add X_1+.....+ X_8 = T
as we wish to total cations to 24 then multiply each of X_1, X_2....., X_8 by 24 and divide by T.

Result = Y_1, Y_2,..... Y_8.

Step 3. $(MgO+MnO+NiO+CaO+Fe^{2+})$ = 8
and
$(Cr+Al+Ti+Fe^{3+})$ = 16
Add $(Y_7+Y_5+Y_6+Y_8+Y_4)$ = α

Subtract 8 from α (=β); if difference +ve continue, if difference -ve stop with message.

Step 4. Check: (Cr+Al+Ti+β) = 16± rounding errors (if any).

Step 5. Calculate wt.% Fe_2O_3
Take β from Step 3, divide by 24 and multiply by T from Step 2 = γ (= number of cations in mol proportion).
Then wt.% = γ x 159.70 ÷ 2 = ξ.

Step 6. Calculate new value of FeO.
Take Y_4 (from Step 2), subtract β (from Step 3) = λ.
Calculate number of cations in mol proportion = λ ÷ 24 x T (Step 2) = δ.
Thus new wt.% of FeO = δ x 71.85 = ε.

Step 7. Normalise wt.% with new values for FeO and Fe_2O_3.
Take value of TiO_2, Al_2O_3, Cr_2O_3, MnO, NiO, MgO and CaO from input, add wt.% Fe_2O_3 (=ξ) plus new wt.% of FeO (=ε).
add = T_2

→ normalisation factor = $\frac{100.00}{T_2}$ = T_3

Multiply each oxide by T_3.

OUTPUT

(1) Print input.

(2) Print normalised values from Step 6.

(3) Print structural formula.

Form:	*Input*	*Step 6*	*Structural*
TiO_2			
Al_2O_3			
Cr_2O_3			
Fe_2O_3			
FeO			
MnO			
NiO			
MgO			
CaO			

Structural formula obtained from:

TiO_2	=	Y_1
Al_2O_3	=	Y_2
Cr_2O_3	=	Y_3
Fe_2O_3	=	β
FeO	=	λ
MnO	=	Y_5
NiO	=	Y_6
MgO	=	Y_7
CaO	=	Y_8

APPENDIX 2

Example of interactive run

```
*SPINEL
  ENTER SAMPLE IDENTIFICATION
>TEST1
  ENTER WEIGHT PERCENT OF  TiO2
>0.0
  ENTER WEIGHT PERCENT OF Al2O3
>6.32
  ENTER WEIGHT PERCENT OF Cr2O3
>57.74
  ENTER WEIGHT PERCENT OF   FeO
>29.66
  ENTER WEIGHT PERCENT OF   MnO
>0.54
  ENTER WEIGHT PERCENT OF   NiO
>0.0
  ENTER WEIGHT PERCENT OF   MgO
>4.24
  ENTER WEIGHT PERCENT OF   CaO
>0.0
  AGAIN?            (N/Y)
>Y
  ENTER SAMPLE IDENTIFICATION
>TEST2
  ENTER WEIGHT PERCENT OF  TiO2
>0.66
  ENTER WEIGHT PERCENT OF Al2O3
>5.06
  ENTER WEIGHT PERCENT OF Cr2O3
>34.41
  ENTER WEIGHT PERCENT OF   FeO
>51.05
  ENTER WEIGHT PERCENT OF   MnO
>0.43
  ENTER WEIGHT PERCENT OF   NiO
>0.0
  ENTER WEIGHT PERCENT OF   MgO
>4.64
  ENTER WEIGHT PERCENT OF   CaO
>0.0
  AGAIN?            (N/Y)
>N
 STOP
```

PRINTED OUTPUT FROM INTERACTIVE RUN EXAMPLE

SAMPLE TEST1

TiO2	0.0000	0.0000	0.0000
Al2O3	6.3200	2.1117	6.3073
Cr2O3	57.7400	12.9373	58.3546
Fe2O3	---	0.9510	4.5063
FeO	29.6600	6.0794	25.9209
MnO	0.5400	0.1296	0.5457
NiO	0.0000	0.0000	0.0000
MgO	4.2400	1.7909	4.2851
CaO	0.0000	0.0000	0.0000
SUMME	98.5000	24.0000	99.9999

Mg/(Mg+Fe) is 22.8%

SAMPLE TEST2

TiO2	0.6600	0.1408	0.6603
Al2O3	5.8600	1.9603	5.8630
Cr2O3	34.4100	7.7188	34.4273
Fe2O3	---	6.1001	28.9566
FeO	51.0500	5.9345	25.0202
MnO	0.4300	0.1034	0.4302
NiO	0.0000	0.0000	0.0000
MgO	4.6400	1.9622	4.6423
CaO	0.0000	0.0000	0.0000
SUMME	97.0500	24.0000	100.0000

Mg/(Mg+Fe) is 24.8%

APPENDIX 3

Program SPINEL

```
C * * * * * * * * * * * * * * * * * * *
      REAL X(8),Y(10),Z(10),LAMBDA,Q(10)
      CHARACTER*12 SAMP
      CHARACTER*5 ELEM(10)
      CHARACTER*1 ANS
      LOGICAL AGAIN
      DATA AGAIN /.FALSE./
      DATA ELEM /' TiO2','Al2O3','Cr2O3','  FeO','  MnO','  NiO',
     1'  MgO','  CaO','Fe2O3','SUMME'/
   10 SAMP='            '
      Z(10)=0.0
      Y(10)=0.0
      Q(10)=0.0
      IF(AGAIN) THEN
      DO 20 J=4,8
      ELEM(J)=ELEM(J+1)
   20 CONTINUE
      ELEM(9)='Fe2O3'
      ENDIF
      WRITE(5,11)
   11 FORMAT(1X,'ENTER SAMPLE IDENTIFICATION')
      READ(5,13) SAMP
   13 FORMAT(A12)
      DO 100 J=1,8
      WRITE(5,40) ELEM(J)
   40 FORMAT(1X,'ENTER WEIGHT PERCENT OF ',A5)
      READ(5,*,ERR=700) Z(J)
      Z(10)=Z(10)+Z(J)
  100 CONTINUE
C * * * * * STEP 1 * * * * * * * * *
      X(1)=Z(1)/79.9
      X(2)=2.0*Z(2)/101.94
      X(3)=2.0*Z(3)/152.02
      X(4)=Z(4)/71.85
      X(5)=Z(5)/70.94
      X(6)=Z(6)/74.71
      X(7)=Z(7)/40.32
      X(8)=Z(8)/56.08
C * * * * * * STEP 2  * * * * * * * * *
      T=X(1)+X(2)+X(3)+X(4)+X(5)+X(6)+X(7)+X(8)
      DO 125 J=1,8
      Y(J)=X(J)*24.0/T
  125 CONTINUE
C * * * * * * STEP 3 * * * * * * * * * *
      ALPHA=Y(7)+Y(5)+Y(6)+Y(8)+Y(4)
      BETA=ALPHA-8.0
      IF(BETA.LT.0.0) THEN
      WRITE(5,160) BETA
      WRITE(6,555) SAMP
      WRITE(6,160) BETA
  160 FORMAT(1X,'BETA =',F6.2,'  . . . . BAD ANALYSIS')
      AGAIN=.FALSE.
      GO TO 10
      ENDIF
```

```
C * * * * * * STEP 4 * * * * * * * * * * **
      IF(ABS(Y(1)+Y(2)+Y(3)+BETA-16.0).GT.0.0001) THEN
      WRITE(5,180) (Y(1)+Y(2)+Y(3)+BETA)
  180 FORMAT(1X,'Al+++, Ti++++, Cr+++, BETA SUM IS ',F9.5)
      WRITE(5,182)
  182 FORMAT(1X,'DO YOU WISH TO CONTINUE?    (N/Y)')
  181 READ(5,183) ANS
  183 FORMAT(A1)
      IF(ANS.EQ.'N') THEN
      WRITE(6,555) SAMP
      WRITE(6,180)  (Y(1)+Y(2)+Y(3)+BETA)
      AGAIN=.FALSE.
      GO TO 10
      ENDIF
      IF(ANS.NE.'Y') THEN
      WRITE(5,184)
  184 FORMAT(1X,'ANSWER N OR Y')
      GO TO 181
      ENDIF
      ENDIF
C * * * * * * STEP 5   Fe2O3 * * * * * * * * * *
      GAMMA=T*BETA/24.0
      PSI=GAMMA*159.70/2.0
C * * * * * * STEP 6   FeO * * * *** **  *  *
      LAMBDA=Y(4)-BETA
      IF(LAMBDA.LT.0.0) THEN
      WRITE(6,555) SAMP
      WRITE(5,204)
      WRITE(6,204)
  204 FORMAT(1X,'RATIO OF 2+:3+ > 1:2 . . . .NOT A SPINEL! TRY AGAIN')
      AGAIN=.FALSE.
      GO TO 10
      ENDIF
      DELTA=T*LAMBDA/24.0
      EPSILO=DELTA*71.85
C * * * * * * * * STEP 7 * * * * * * * * * * * *
      T2=Z(1)+Z(2)+Z(3)+EPSILO+PSI+Z(5)+Z(6)+Z(7)+Z(8)
      T3=100.0/T2
      Q(5)=EPSILO*T3
      Q(4)=PSI*T3
      DO 400 J=1,3
      Q(J)=Z(J)*T3
  400 CONTINUE
      DO 410 J=6,9
      Q(J)=Z(J-1)*T3
  410 CONTINUE
      DO 420 J=9,5,-1
      ELEM(J)=ELEM(J-1)
      Z(J)=Z(J-1)
      Y(J)=Y(J-1)
  420 CONTINUE
      ELEM(4)='Fe2O3'
      Y(4)=BETA
      Y(5)=LAMBDA
      DO 450 J=1,9
      Q(10)=Q(10)+Q(J)
      Y(10)=Y(10)+Y(J)
  450 CONTINUE
```

```
C * * * * * * * PRINT OUTPUT * * * * * * * * * *
      WRITE(6,555) SAMP
  555 FORMAT(////,1X,'SAMPLE  ',A12,/)
      DO 600 J=1,10
      IF(J.EQ.4) THEN
      WRITE(6,585) ELEM(J),Y(J),Q(J)
  585 FORMAT(1X,A5,'        ---   ',2F12.4)
      ELSE
      WRITE(6,590) ELEM(J),Z(J),Y(J),Q(J)
  590 FORMAT(1X,A5,3F12.4)
      ENDIF
  600 CONTINUE
      WRITE(6,606) (Y(8)/(Y(8)+Y(5)))*100.0
  606 FORMAT(/,1X,'Mg/(Mg+Fe) is ',F4.1,'%')
      WRITE(5,680)
  680 FORMAT(1X,'AGAIN?              (N/Y)')
  682 READ(5,183) ANS
      IF(ANS.EQ.'Y') THEN
      AGAIN=.TRUE.
      GO TO 10
      ENDIF
      IF(ANS.EQ.'N') GO TO 900
      WRITE(5,684)
  684 FORMAT(1X,'ANSWER N OR Y')
      GO TO 682
  700 WRITE(5,701)
  701 FORMAT(1X,'DATA INPUT ERROR . . .  START AGAIN')
      GO TO 10
  900 WRITE(6,901)
  901 FORMAT(/)
      STOP
      END
```